Determinare le CI che compensano ESATTAMENTE la risposta transitoria nel caso di ingresso a gradino unitario per un Sistema LTI-TC

In[•]:= A = {{0, 1/2, 0, 1/2}, {-10, -11, 14, 9}, {-10, -11, 14, 10}, {8, 9, -12, -9}}; B = {{0}, {1}, {1}, {-1}}; C1 = {{1, -1/2, 0, -1/2}};

In[•]:=
```
Eigenvalues[A]
```

Out[•]=
```
{-2, -2, -1, -1}
```

Vado ad “incapsulare” la terna A,B,C in una struttura StateSpaceModel

In[•]:=
```
Σ = StateSpaceModel[{A, B, C1}]
```

Out[•]=

$$\left(\begin{array}{cccc|c} 0 & \frac{1}{2} & 0 & \frac{1}{2} & 0 \\ -10 & -11 & 14 & 9 & 1 \\ -10 & -11 & 14 & 10 & 1 \\ 8 & 9 & -12 & -9 & -1 \\ \hline 1 & -\frac{1}{2} & 0 & -\frac{1}{2} & 0 \end{array}\right)^{\mathcal{S}}$$

Calcolo la funzione di trasferimento

In[•]:=
```
G[s_] := Simplify[C1.Inverse[s IdentityMatrix[4] - A].B]〚1〛〚1〛
```

In[•]:=
```
G[s]
```

Out[•]=

$$\frac{-1+s}{\left(2+3\,s+s^2\right)^2}$$

In[•]:=
```
Eigenvalues[A]
```

Out[•]=
```
{-2, -2, -1, -1}
```

I modi naturali del sistema sono exp(-t), t exp(-t), exp(-2 t), t exp(-2 t)

In[∘]:=
```
JordanDecomposition[A][[2]] // MatrixForm
```

Out[∘]//MatrixForm=

$$\begin{pmatrix} -2 & 1 & 0 & 0 \\ 0 & -2 & 0 & 0 \\ 0 & 0 & -1 & 1 \\ 0 & 0 & 0 & -1 \end{pmatrix}$$

Scrivo il vettore dello stato iniziale

In[∘]:=
```
x0 = {{x1}, {x2}, {x3}, {x4}}
```

Out[∘]=

$\{\{x_1\}, \{x_2\}, \{x_3\}, \{x_4\}\}$

Associo ad una variabile di Mathematica la risposta libera in s

In[∘]:=
```
libera = Simplify[C1.Inverse[s IdentityMatrix[4] - A].x0][[1]][[1]]
```

Out[∘]=

$$\left(2\left(10+18\,s+7\,s^2+s^3\right)x_1-(-1+s)\left(s\,(4+s)\,x_2+2\,(6+s)\,x_3+\left(14+6\,s+s^2\right)x_4\right)\right)\Big/\left(2\left(2+3\,s+s^2\right)^2\right)$$

Associo ad una variabile di Mathematica la risposta al gradino unitario

In[∘]:=
```
forzata = Simplify[G[s] (1/s)]
```

Out[∘]=

$$\frac{-1+s}{s\left(2+3\,s+s^2\right)^2}$$

Associo ad una variabile di Mathematica la risposta a regime (in questo caso, gradino unitario)

In[∘]:=
```
regime = G[0] (1/s)
```

Out[∘]=

$$-\frac{1}{4\,s}$$

In[∘]:=
```
transitoria = Factor[forzata - regime]
```

Out[∘]=

$$\frac{16+13\,s+6\,s^2+s^3}{4\,(1+s)^2\,(2+s)^2}$$

Sommo libera e transitoria e ne estraggo il numeratore (una frazione e' nulla se nullo e' il suo numeratore)

In[∘]:=
```
Numerator[Simplify[Expand[libera + transitoria]]]
```

Out[∘]=

$$16+13\,s+6\,s^2+s^3+4\left(10+18\,s+7\,s^2+s^3\right)x_1-2\,s\left(-4+3\,s+s^2\right)x_2+24\,x_3-20\,s\,x_3-4\,s^2\,x_3+28\,x_4-16\,s\,x_4-10\,s^2\,x_4-2\,s^3\,x_4$$

Estraggo i coefficienti del polinomio al numeratore (libera+forzata)

In[•]:=
```
CoefficientList[Numerator[Simplify[Expand[libera + transitoria]]], s]
```

Out[•]=

$$\{16 + 40\,x_1 + 24\,x_3 + 28\,x_4,\ 13 + 72\,x_1 + 8\,x_2 - 20\,x_3 - 16\,x_4,\ 6 + 28\,x_1 - 6\,x_2 - 4\,x_3 - 10\,x_4,\ 1 + 4\,x_1 - 2\,x_2 - 2\,x_4\}$$

Determino ora le incognite xi che annullano il set dei coefficienti del polinomio a numeratore di libera+forzata

In[•]:=
```
Solve[CoefficientList[Numerator[Simplify[Expand[libera + transitoria]]], s] ==
  {0, 0, 0, 0}, {x1, x2, x3, x4}]
```

Out[•]=

$$\left\{\left\{x_1 \to -\frac{1}{4},\ x_2 \to 0,\ x_3 \to -\frac{1}{4},\ x_4 \to 0\right\}\right\}$$

In[•]:=
```
regime
```

Out[•]=

$$-\frac{1}{4\,s}$$

Prova del nove, calcolo della risposta a partire dallo stato iniziale cosi’ determinato

In[•]:=
```
OutputResponse[{Σ, {-1 / 4, 0, -1 / 4, 0}}, 1, t]
```

Out[•]=

$$\left\{-\frac{1}{4}\right\}$$